Markov: Un ejemplo sencillo

Vamos a calcular la evolución de las distribuciones de probabilidad $\mu^{(n)}$ a partir de una distribución inicial conocida $\mu^{(0)}$ y una matriz de transiciones de probabilidad P dadas, en un caso sencillo. Concretamente, consideramos el caso de una cadena de Markov $\{X_n\}_{n=0}^{\infty}$ con solo dos estados y distribución inicial $\mu^{(0)} = (1/2, 1/2)$. La matriz de transiciones viene dada por

$$\begin{matrix} 0.175 & 0.825 \\ 0.526 & 0.474 \end{matrix}$$

Como primer paso, calculamos los autovalores y autovectores de P. De este modo, podremos comprobar que P es diagonalisable y obtener las matrices D y Q tales que

$$P = QDQ^{-1},$$

donde D es diagonal. Eso conduce luego a que

$$P^n = QD^nQ^{-1},$$

siendo D^n muy fácil de calcular. Es más, esto nos permite un cálculo explícito de P^n y, por tanto, de

$$\mu^{(n)} = \mu^{(0)}P^n, \;\; n = 0, 1, 2, \cdots$$

```
P = matrix(QQ ,[[0.175 , 0.825],[0.526 ,0.474]])
show("P=",P," Valores propios:", P.eigenvalues())
p=P.characteristic_polynomial(); show("p(x)=  ",p)
```

P= $\begin{pmatrix} \frac{7}{40} & \frac{33}{40} \\ \frac{263}{500} & \frac{237}{500} \end{pmatrix}$ Valores propios: $\left[1, -\frac{351}{1000}\right]$

p(x)= $x^2 - \frac{649}{1000}x - \frac{351}{1000}$

```
D = diagonal_matrix([P.eigenvalues()[0],
P.eigenvalues()[1]])
show("D=", D)
```

D= $\begin{pmatrix} 1 & 0 \\ 0 & -\frac{351}{1000} \end{pmatrix}$

```
P.eigenvectors_right()
```

```
[(1, [
  (1, 1)
  ], 1), (-351/1000, [
  (1, -526/825)
  ], 1)]
```

```
q1=(P.eigenvectors_right()[0])[1][0];
q2=(P.eigenvectors_right()[1])[1][0];
show("q1=",q1," q2=",q2)
```

q1= $(1, 1)$ q2= $\left(1, -\frac{526}{825}\right)$

```
q1
```

```
(1, 1)
```

```
Q=column_matrix([q1,q2]);
show("Q=",Q)
print P == Q*D*Q.inverse()
```

Q= $\begin{pmatrix} 1 & 1 \\ 1 & -\frac{526}{825} \end{pmatrix}$

```
True
```

Ahora calculamos una expresión explícita para P^n, lo que luego nos permite dibujar $\mu^{(n)}$ para n=0,1,2,\... como un par de sucesiones (la primera componente, en rojo, la segunda en azul) en un mismo gráfico.

```
mo = vector((1/2,1/2));
var('n')
D^n
show( mo*Q*D^n*Q.inverse())
```

$\left(\frac{299\,(-351)^n}{2702 \cdot 1000^n} + \frac{526}{1351}, -\frac{299\,(-351)^n}{2702 \cdot 1000^n} + \frac{825}{1351}\right)$

```
list_plot([(k,299*(-351)^k/(2702*1000^k)+526/1351) for k in
range(1,50)],color='red')+list_plot([(k,-299*
(-351)^k/(2702*1000^k)+825/1351) for k in range(1,50)])
```



ESTÁ CLARO QUE LA DISTRIBUCIÓN DE PROBABILIDADES DE X_n ES ESTABILIZA PARA n GRANDE...